

## Medieaftale og mediecheck

### Resumé:

*Medieaftalen fra 2018 udfaser licensen og ændrer mediestøtten sådan at Danmarks Radios aktivitet beskæres med ca. 20 pct. og at private aktører får bedre mulighed for at bidrage. Public service opgaverne finansieres fremover af indkomstkatten. Medieaftalen gennemføres over 5 år i perioden 2019-2024.*

*Papiret beskriver hvordan aftalen kan indarbejdes i fremskrivninger med Adam. Herunder foreslås at ligningerne tilpasses sådan at udgiften til mediechecken bliver nemmere at håndtere.*

---

Nøgleord: Medieaftale, licens, multimedieskat, mediecheck, indkomstskat

*Modelgruppepapirer er interne arbejdspapirer. De konklusioner, der drages i papirerne, er ikke endelige og kan være ændret inden opstillingen af nye modelversioner. Det henstilles derfor, at der kun citeres fra modelgruppepapirerne efter aftale med Danmarks Statistik.*

## 1 Indledning

I 2018 indgår regeringen (Venstre, Liberal Alliance og Det Konservative Folkeparti) og Dansk Folkeparti en aftale om public service opgaver og støtten til danske medier. Aftalen indebærer at Danmarks Radios aktiviteter beskæres og andre aktører får en større del af mediestøtten.

Medieaftalen ændrer samtidig finansieringen af public serviceopgaverne og mediestøtten. Licensen og medieskatten afskaffes i aftaleperioden, og opgaverne finansieres i stedet over skatten. Aftalen gælder perioden 2019-2023, og indeholder også en mediecheck som forventes at blive en social ydelse for personer som har betalt nedsat licens.

## 2 Licens og moms

I 2018 er licensen 2527 kr. for en familie med fuld licensbetaling og 1263,50 kr. for en familie med nedsat licens. Heraf er moms en del af beløbet. Eksklusiv moms er prisen 2021,60 kr. og med nedsat licens 1010,80 kr. De samlede licensbetalinger udgør ca. 4.5 mia. kr. i 2018. Medtages afledte effekter på momsprovenuet af provenuet er beløbet på ca. 5.6 mia.kr.

Licensbetalingen opfattes i nationalregnskabet som en indkomstskat, og i Adam ligninger er provenuet af licensen beskrevet i variabelen  $Sym$ :

$$(1) \quad Sym = ksym \cdot 0.001 \cdot tsym \cdot ptyl \cdot U1899$$

Ligningen kan opfattes som årlig sats,  $tsym \cdot ptyl$ , der betales af et antal husstande,  $ksym \cdot U1899$ . Den årlige satsen er produktet af licensen  $tsym$ , og reguleringsindekset  $ptyl$ <sup>1</sup>. K-faktoren,  $ksym$ , udtrykker et gennemsnitlig antal personer i de licensbetalende husstande, hvor personer med nedsat licens indgår med mindre vægt.

**Table 1.** Medielicens 2015-2019. Dekomponering ved ligning (1)

	Provenu ekskl. moms $Sym$	Licens ekskl. moms $tsym \cdot ptyl$	Husstande $ksym \cdot U1899$
2015	4477	1968	2275
2016	4411	1981	2227
2017	4441	1994	2228
2018	4503	2022	2227
2019	3302	1572	2101

Tabellen viser at den årlige licenssats har været ca. 2000 kr. frem til 2018, og at licensen betales af ca. 2.2 mio. husstande (omregnet til husstande med fuld licens). I 2019 falder satsen med ca. 25 pct. Faldet er helt i overensstemmelse med at licensen udfases over 4


<sup>1</sup> Mere info om reguleringsindekset kan findes i TMK030619.

år. Antallet af husstande falder også lidt. Der kan være flere forklaringer. Det kan tænkes at opgørelsen af provenuet er undervurderet. Opgørelsen er foreløbig og kan blive revideret. Andre faktorer kan spille en rolle. Fx vil ændringen af pensionsalderen påvirke sammensætningen af licensbetalere, sådan at andelen med fuld licens stiger.

Tallene fra 2019 antyder at licensen aftrappes lineært med en fjerdedel om året i fire år. Udfasningen af licensen kan beskrives ved at aftrappe satsen i perioden 2019-2022.

SERIES <2019 2022> tsym = 1306, 870, 435, 0 ; // tsym(2018)=1741

**Figur 1.** Fordeling af husholdningernes licensbetaling på kr. 2527 i 2018 (Kilde Danmarks Radio Årsrapport 2018)



Afskaffelsen af licensen giver samtidig et tab af momsprovenu, som kan skønnes at være 1126 mio. kr. i 2018-priser. Det fremgår ikke klart hvordan moms, som opkræves i forbindelse med licensbetaling, indgår i nationalregnskabet<sup>2</sup>. I dette regneeksempel antages forenkende at moms betales af erhvervet af øvrige tjenester:

$$(2) \quad Spg_{xqz} = btgxqz \cdot tg \cdot Vqz / (1 + btgxqz \cdot tg)$$

Momsprovenuet,  $Spg_{xqz}$ , kædes sammen med materialekøbet,  $Vqz$ , og momssatsen,  $tg$ . Men nogle virksomheder er fritaget for moms, og relationen indeholder derfor også en momsbelastningsgrad,  $btgxqz$ . Hvis branchen har en stor andel af momsfrie aktiviteter, er momsbelastningsgraden lille.

Nedenfor ændres momsbelastningsgraden gradvist sådan at det svarer til et provenutab på 1126 mio.kr.

<sup>2</sup> Hvis nationalregnskabet kan bidrage med oplysninger om momsafregningen af licensen, kan den simple beregning af effekten på momsprovenuet forbedres og blive mere realistisk.

```

CREATE xspg_xqz;
SERIES <2019 2022> xspg_xqz = 281, 563, 845, 1126 ;
SERIES <2019 2022> btgxqz = btgxqz - xspg_xqz / (tg*(Vqz + xspg_xqz)) ;
DELETE xspg_xqz ;

```

### 3 Indkomstskat

Medieaftalen indeholder en forøgelse af indkomstskatten. Instrumentet er en nedsættelse af personfradraget med 2900 kr. Bundfradrag i indkomstskatten kan analyseres i Pskat. Pskat omregner ændringer i bundfradraget til ændringer i andelen af skattepligtig og personlig indkomst, der beskattes med regelgivne skattesatser.

For en gennemsnitslig skatteborger vil en reduktionen af personfradraget på 2900 kr. øge andelen af personlig indkomst, som beskattes med bundskat, med 0.9 pct. point. For kommuneskatten øges skattegrundlaget i gennemsnit med 1.pct.point. For skatteydere med høje indkomster er effekten på skattegrundlaget mindre end gennemsnittet. For skatteydere med lave indkomster er effekten lidt større.

Pskat-kørslen giver følgende output:

```

SERIES <2019 2022> bysp10s + $ 0.000939 0.001838 0.002683 0.003598 ;
SERIES <2019 2022> bysp10w + $ 0.001833 0.003586 0.005233 0.007016 ;
SERIES <2019 2022> bysp10l + $ 0.003504 0.006857 0.010006 0.013417 ;
SERIES <2019 2022> bysp10ef + $ 0.003037 0.005942 0.008671 0.011625 ;
SERIES <2019 2022> bysp10fp + $ 0.003123 0.006113 0.008920 0.011962 ;
SERIES <2019 2022> bysp10q + $ 0.004114 0.008070 0.011799 0.015849 ;

SERIES <2019 2022> bys10s + $ 0.002051 0.004018 0.005869 0.007875 ;
SERIES <2019 2022> bys10w + $ 0.002512 0.004918 0.007178 0.009626 ;
SERIES <2019 2022> bys10l + $ 0.004059 0.007942 0.011590 0.015539 ;
SERIES <2019 2022> bys10ef + $ 0.003266 0.006392 0.009328 0.012506 ;
SERIES <2019 2022> bys10fp + $ 0.003159 0.006184 0.009026 0.012103 ;
SERIES <2019 2022> bys10q + $ 0.004331 0.008490 0.012400 0.016657 ;

```

For en skatteyder med en skattepligtig indkomst på 290.000 kr. betyder en ændring i personfradraget på 2.900 kr. at basen for skatten ændres 1 pct. point. Beregningen i Pskat illustrerer at lønmodtagernes gennemsnitlig skattepligtig indkomst er en anelse større end 290.000 kr. Selvstændige ligger i gennemsnit lidt højere, mens de øvrige socioøkonomiske grupper har en lavere indkomst i gennemsnit.

### 4 Mediecheck

Omlægning fra licens til indkomstskat har fordelingsmæssige effekter. Når en skat flytter fra husstande til personer, så er det en skattelettelse for enlige og en skattestigning for par. Finansministeriet har udarbejdet et fakta ark med forventede fordelingsmæssige effekter. Her fremgår følgende fordelingseffekter

**Table 2.** Effekt på indkomst efter indfasning (2018-priser) – udvalgte familietyper

	Licens	Personfradrag	Mediecheck	I alt
Enlige				
LO-arbejde	2343	-1101	0	1242
Pensionist fuld licens	2343	-1101	0	1242
..Pensionist nedsat licens	1171	-1101	558	628
Par				
LO-arbejde	2343	-2177	0	166
Pensionist fuld licens	2343	-2177	0	166
..Pensionist nedsat licens	1171	-2177	1124	118

Kilde: Finaministeriet, Fakta ark om afskaffelse af licens – virkning for familietyper

I den samlede medieaftale er der en positiv effekt på disponibel indkomst for alle udvalgte familietyper. Det hænger sammen med at aftalen ikke kun handler om at flytte finansiering. Sammen med afskaffelsen af licensen gennemføres også besparelser i mediestøtten, således at kravet til finansiering bliver mindre, og den negative effekt fra skatteværdien af mistet bundfradrag kommer ikke til at udligne gevinsten ved bortfaldet af licensbetalingen. Samlet set er effekten for par således positiv, men lille. For enlige er der en større positiv effekt på omtrent 100 kr. om måneden eller 1200 kroner om året. På længere sigt kan effekten være mindre. Bundfradraget reguleres med lønudviklingen, mens licensen ikke er steget i takt med lønnen i de seneste år.

Personer uden indkomst eller lille indkomst slipper helt for skatten, mens alle skatteydere, som betaler kommune- og bundskat, kommer til at betale det samme beløb i ekstra skat. Det betyder umiddelbart at skatten stiger mere end licensen falder for mange pensionister. Mange pensionister betaler nedsat licens. Det er formentlig baggrunden for at medieaftalen omfatter en mediecheck for personkredsen som har betalt nedsat licens. Efter indfasning er mediechecken på 900 kr. årligt til den personkreds, der har betalt nedsat licens.

Mediecheck kan sidestilles med den grønne check. I nationalregnskabet er den grønne check ikke et nedslag i skatten. Checken er en social ydelse. Mediechecken må også forventes at blive en social ydelse.

I denne sammenhæng er det nemmest at udvide antallet af indkomstoverførsler

$$(3) \quad Tyr = Tyrk + Tyrbf + Tyr gc + \mathbf{Tyr mc} + Tyrh + Tyr r$$

Hvor  $Tyr mc$  er mediechecken. Relationen for mediechecken tager udgang på i satsen  $ttyr mc$ , som efter indfasning er 900 kr. i 2018-priser omregnet til 2010 priser. Personkredsen er folkepensionister og førtidspensionister,  $Upfpi + Upfoi$ .

$$(4) \quad Tyr mc = ktyr mc \cdot 0.001 \cdot ttyr mc \cdot pttyl \cdot (Upfpi + Upfoi)$$

Ligning (4) har også en k-faktor,  $ktyr mc$ , som afspejler at det ikke er alle folke- og førtidspensionister, der får tildelt en mediecheck. I 2019 udbetales 116 mio. kr. til modtagere af mediechecken. Det giver  $ktyr mc$  værdien 0.3847;

Hvis vi antager at mediechecken indfases i perioden 2019-2022, fås

```
// ttyr mc(2022)=900/pttyl(2018)
SERIES <2019 2022> ttyr mc = 194, 388, 581, 775;
```

Bemærk at skattepligtige indkomstoverførsler er beskrevet ved variabelen,  $Tys$ , hvor alle ikke-skattepligtige variabler nævnes eksplicit

$$(5) \quad Tys = Ty_{o_h} - (Tyrkrr + Tyrbf + Tyr gc + Tyrh + Tyrrr)$$

Mediechecken er skattepligtig, så der er ingen grund til at ændre  $Tys$ - relationen. Men mediechecken skal tælles med i den rigtige indkomstgruppe i skattefunktionen. Mediechecken kan lægges til personlig indkomst for pensionister. Dvs.

$$(6) \quad Yasp = k yas \cdot (Typfp + Typr + Tyrhy + Tyrmc) \\ + (1 - bsyptyr_{bf}) \cdot (Typcr_{bf} + Typir_{bf}) \\ + Typcr_{dmp} \\ + (1 - bsyptypcr_{sp}) \cdot Typcr_{sp} \\ + (1 - bsyptypcr_{atp}) \cdot Typcr_{atp}$$

I modelversionen jun19 er det ikke muligt at tilføje en ekstra social overførsel. Her bruges ovenstående beregning til at justere i udbetalinger af øvrige skattepligtige overførsler,  $Tyrrs$ . Regneeksemplet til modelversionen jun19 findes i bilag 3.

## 5 Udgiften til public serviceopgaver sænkes

Den samlede medieaftale har følgende ramme:

Økonomi i medieaftale

Mio.kr. (2018-niveau)	2018	2019	2020	2021	2022	2023
Trykte og skrevne internetbaserede medier	410,7	410,7	429,4	435,6	442,9	451,8
-heraf pulje til trykte medier, blade og tidsskrifter og innovation	404,7	404,7	359,2	359,2	359,2	359,2
-heraf pulje til internetmedier	6,0	6,0	45,2	51,4	58,7	67,6
-heraf pulje til distrikts- og ugeaviser	-	-	25,0	25,0	25,0	25,0
DR	3.767,5	3.591,8	3.459,1	3.347,7	3.229,7	3.092,5
De regionale TV 2-virksomheder	526,1	526,1	526,1	534,1	526,1	526,1
Lokalradio- og tv	51,3	52,3	52,3	52,3	52,3	52,3
-heraf til SAMLtil fælles lokalradio- og tv-formål	-	1,0	1,0	1,0	1,0	1,0
Filmstøtteordninger	25,0	145,0	145,0	145,0	145,0	145,0
-heraf til pulje til regionale filmfonde udenfor hovedstaden	-	35,0	35,0	35,0	35,0	35,0
Public Service-Pulje	35,0	68,8	62,6	54,0	62,6	100,6
Driftstilskud til FM 4 iht tilladelse	95,3	94,8	92,4	89,6	86,7	83,9
Ny DAB radiokanal (udbud)	-	-	30,0	30,0	30,0	30,0
Ny PS-tv-kanal (udbud)	-	-	50,0	50,0	50,0	50,0
Arkivadgang til DR-udsendelser	-	-	14,0	8,0	8,0	8,0
Overførsel af tilskud til Station Next til filmforhandlinger	6,9	6,9	6,9	6,9	6,9	6,9
Markedstest af nye initiativer	2,0	2,0	2,0	2,0	2,0	2,0
Årlig rapportering om medieudviklingen, kulturstatistik mv.	8,7	8,7	8,7	8,7	8,7	8,7
Til disposition	7,2	-	-	-	-	-
<b>I alt</b>	<b>4.935,8</b>	<b>4.907,1</b>	<b>4.878,5</b>	<b>4.763,8</b>	<b>4.650,9</b>	<b>4.557,7</b>

Note: Kulturministeren fastlægger for årene 2019 -2021 den nærmere fordeling af de i tabellen angivne beløb på andele af licensprovenu hhv. tilskud over finansloven.

Det ligger i aftalen at udgifterne falder 388 mio. kr. i perioden frem mod 2025. Den umiddelbare besparelse på DR's budgetter er ca. 700 mio. kr. Det modsvares af andre ekstra udgifter på 300 mio. kr.

DR's virksomhed opfattes i nationalregnskabet, som produktion af offentlige tjenesteydelser. I dette regneeksempel ligges besparelsen på 388 mio.kr. ind som en nedgang i offentlige beskæftigelse, produktion og forbrug.

```
CREATE besparelse ;  
SERIES <2019 2023> besparelse = 78, 155, 233, 310, 388 ;  
SERIES zQo1 = zQo1 + (@Ywo1/@Xo1)*besparelse/(lnakk*hgo1*0.001*klo1) ;  
DELETE besparelse ;
```

Lønudgifterne udgør ca. 56 pct. ( $=@Ywo1/@Xo1$ ) af de samlede produktionsudgifter. Modellens sammenhænge sikrer at varekøb og produktionsafgifter øges svarende til ændringen i lønomkostningerne.

En grundigere gennemgang kan tage højde for alle bruttobevægelser i udgiftsrammen ovenfor. Her vil besparelsen på DR reducere offentlig produktion med ca. 700 mio. kr. De øgede udgifter på ca. 300 mio. kr. ville være fordelt på produktionsomkostninger, subsidier og tilskud. Effekten på de samlede offentlige finanser ville være den samme, men fordelingen på offentlige forbrug og andre udgifter vil blive lidt anderledes.

## 6 Opsamling

Ovenstående medtages i modelgruppens fremskrivningsmodul i januar 2020.

Den marginale effekt af medieaftalen er ekspansiv. Der hænger sammen med at mediepakken er underfinansieret. Finansieringen svarer til udgifterne i den økonomiske ramme. Men licensbetalingerne var momsbelagte. Derfor svarer ophævelsen af licensen til en skattelettelse på omtrent 20 pct. af den oprindelse licens, når aftalen er indfaset i 2025.

## Bilag 1. Forslag til ændringer i model og data

Forslag til ændrede ligninger. Ændringer er angivet med fed skrift. Hvis ligningerne bruges i Adam jun19, så kaldes modelversionen Jun19x i ovenstående.

```
FRML _GJ_D Tyrmc = ktyrmc*0.001*ttyrmc*pttyl*(Upfpi+Upfoi) $      () Mediecheck
FRML _D   Tyr   = Tyrk + Tyrbf + Tyrge + Tyrmc + Tyrh + Tyrp $ () Øvrige i alt
FRML _GJ_D Yasp = kyas*(Typfp+Typr+Tyrhy+Tyrmc
                    +(1-bsyptypr_bf) * (Typcr_bf+Typir_bf)+Typcr_dmp
                    +(1-bsyptypcr_sp)*Typcr_sp+(1-bsyptypcr_atp)*Typcr_atp) $
```



## Bilag 2. Marginale effekter af medieaftale – jun19x

Koden nedenfor bruger modellen Jun19x (fra Bilag 1) og grundforløbet lang100.

```
CLOSE *; CLEAR ; MODE sim;
MODEL jun19x;
READ lang100 ;
SERIES <1966 2110> Tyrmc = 0 ;
SERIES <1966 2110> ttyrmc = 0 ;
SERIES <1966 2110> ktyrmc = 0 ;
TIME 2019 2110 ;
SIM <2019 2110> ;
WRITE grund;

MODEL jun19x;
READ grund;
TIME 2019 2110 ;

// Licens aftrappes 2019-2022
SERIES <2019 2022> tsym = 1306, 870, 435, 0 ; // tsym(2018)=1741
SERIES <2023 2110> tsym = 0;

// Momsprovenue sænkes
CREATE xspg_xqz;
SERIES <2019 2022> xspg_xqz = 281, 563, 845, 1126 ;
SERIES <2019 2022> btgxqz = btgxqz - xspg_xqz / (tg*(Vqz + xspg_xqz)) ;
DELETE xspg_xqz ;
SERIES <2023 2110> btgxqz = btgxqz[2022]*@btgxqz/@btgxqz[2022] ;

// Mediecheck med foreslåede ligninger
// Vi antager at 38 pct. af pensionisterne har nedsat licens
// ttyrmc(2022)=900/pttyl(2018)
SERIES <2019 2022> ttyrmc = 194, 388, 581, 775;
SERIES <2023 2110> ttyrmc = ttyrmc[2022] ;
SERIES <2019 2110> ktyrmc = 0.3847 ;

// Indkomstskat
// Personfradrag sænkes kr 2900
SERIES <2019 2022> bysp10s + $ 0.000939 0.001838 0.002683 0.003598 ;
SERIES <2019 2022> bysp10w + $ 0.001833 0.003586 0.005233 0.007016 ;
SERIES <2019 2022> bysp10l + $ 0.003504 0.006857 0.010006 0.013417 ;
SERIES <2019 2022> bysp10ef + $ 0.003037 0.005942 0.008671 0.011625 ;
SERIES <2019 2022> bysp10fp + $ 0.003123 0.006113 0.008920 0.011962 ;
SERIES <2019 2022> bysp10q + $ 0.004114 0.008070 0.011799 0.015849 ;

SERIES <2019 2022> bys10s + $ 0.002051 0.004018 0.005869 0.007875 ;
SERIES <2019 2022> bys10w + $ 0.002512 0.004918 0.007178 0.009626 ;
SERIES <2019 2022> bys10l + $ 0.004059 0.007942 0.011590 0.015539 ;
SERIES <2019 2022> bys10ef + $ 0.003266 0.006392 0.009328 0.012506 ;
SERIES <2019 2022> bys10fp + $ 0.003159 0.006184 0.009026 0.012103 ;
SERIES <2019 2022> bys10q + $ 0.004331 0.008490 0.012400 0.016657 ;

// Mediestøtten sænkes
CREATE Besparelse ;
SERIES <2019 2023> Besparelse = 78, 155, 233, 310, 388 ;
SERIES <2019 2023> zQo1 = zQo1-Besparelse*(@Ywol/@Xo1)/(lnakk*hgol*0.001*klo1);
DELETE Besparelse ;
SERIES <2024 2110> zQo1 = zQo1[2023] + (@zQo1-@zQo1[2023] ) ;

SIM <2019 2110> ;

WRITE mul;
```

## Bilag 3. Marginale effekter af medieaftale – jun19

Koden nedenfor bruger modellen Jun19 og grundforløbet lang100.

```
MODEL jun19 ;
READ lang100 ;
TIME 2019 2110 ;

// Licens aftrappes 2019-2022
SERIES <2019 2022> tsym = 1306, 870, 435, 0 ; // tsym(2018)=1741
SERIES <2023 2110> tsym = 0;

// Momsprovenuue sænkes
CREATE xspg_xqz;
SERIES <2019 2022> xspg_xqz = 281, 563, 845, 1126 ;
SERIES <2019 2022> btgxqz = btgxqz - xspg_xqz / (tg*(Vqz + xspg_xqz)) ;
DELETE xspg_xqz ;
SERIES <2023 2110> btgxqz = btgxqz[2022]*@btgxqz/@btgxqz[2022] ;

// Mediecheck med foreslåede ligninger
// Vi antager at 38 pct af pensionisterne har nedsat licens
// ttyrmc(2022)=900/pttyl(2018)
CREATE ttyrmc, ktyrmc, Tyrmc;
SERIES <2019 2022> ttyrmc = 194, 388, 581, 775;
SERIES <2023 2110> ttyrmc = ttyrmc[2022] ;
SERIES <2019 2110> ktyrmc = 0.3847 ;
SERIES <2019 2110> Tyrmc = ktyrmc*0.001*ttyrmc*(Upfpi+Upfoi) ;
SERIES <2019 2110> Tyrrsd = @Tyrrsd + Tyrmc;
DELETE ttyrmc, ktyrmc, Tyrmc;

// Indkomstskat
// Personfradrag sænkes kr 2900
SERIES <2019 2022> bysp10s + $ 0.000939 0.001838 0.002683 0.003598 ;
SERIES <2019 2022> bysp10w + $ 0.001833 0.003586 0.005233 0.007016 ;
SERIES <2019 2022> bysp10l + $ 0.003504 0.006857 0.010006 0.013417 ;
SERIES <2019 2022> bysp10ef + $ 0.003037 0.005942 0.008671 0.011625 ;
SERIES <2019 2022> bysp10fp + $ 0.003123 0.006113 0.008920 0.011962 ;
SERIES <2019 2022> bysp10q + $ 0.004114 0.008070 0.011799 0.015849 ;

SERIES <2019 2022> bys10s + $ 0.002051 0.004018 0.005869 0.007875 ;
SERIES <2019 2022> bys10w + $ 0.002512 0.004918 0.007178 0.009626 ;
SERIES <2019 2022> bys10l + $ 0.004059 0.007942 0.011590 0.015539 ;
SERIES <2019 2022> bys10ef + $ 0.003266 0.006392 0.009328 0.012506 ;
SERIES <2019 2022> bys10fp + $ 0.003159 0.006184 0.009026 0.012103 ;
SERIES <2019 2022> bys10q + $ 0.004331 0.008490 0.012400 0.016657 ;

// Mediestøtten sænkes
CREATE Bsparelse ;
SERIES <2019 2023> Bsparelse = 78, 155, 233, 310, 388 ;
SERIES <2019 2023> zQo1 = zQo1-Bsparelse*(@Ywol/@Xo1)/(lnakk*hgo1*0.001*klo1);
DELETE Bsparelse ;
SERIES <2024 2110> zQo1 = zQo1[2023] + (@zQo1-@zQo1[2023] ) ;

SIM <2019 2110> ;

WRITE mul;
```